
websites for FORM

• https://github.com/vermaseren/form


• https://www.nikhef.nl/~form/maindir/documentation/documentation.html

Installation

3



Examples
*** test1.frm

S a,b;

L F = (a+b)^2;

P;

.end

Symbols,a,b; 

Symbols,a b; 

Symbols     ,a        b; 

S a,b;

Sym a b;

sym a b;

sYmB a b;

SYMBOL a,b;

4



*** modify test1.frm 

S a,b;

L F=(a+b)^2;

P;

.end

*** test2.frm

Symbols a,b,c,d,e,f,g;

Local F=(a+b+c+d+e+f+g)^32;

.end

*** Try this in MMA?

*** test3.frm

Symbols a,b,c,d;

Local F = (a+b+c+1)^6;

id a = -c+d+1;

*** .sort

id b=-d+1;

Print;

.end


*** .sort splits module

*** test4.frm

Symbols a,b,c,d;

On HighFirst;

Local F = a+b+c+d;

.sort

id a = (a+b)^2;

id c = b+d;

id b = b+1;

Print;

.end

*** test5.frm

Symbols a,b,c;

Local F = (a+b)^6;

id  a^2*b = c;

Print +s;

.end

Functions
*** test6.frm

Functions A1,B1;

CFunctions A2,B2;

Local F1 = (A1+B1)^3;

Local F2 = (A2+B2)^3;

Print;

.end

*** test7.frm

Functions A1,B1;

CFunctions A2,B2;

Local F1 = (A1+B1+A2+B2)^3;

Local F2 = ((A1+B1)+A2+B2)^3;

Print;

.end

*** test8.frm

Functions A1,B1,C1,D1;

CFunctions A2,B2,C2,D2;

Local F1 = (A1+B1+C1+D1+A2+B2+C2+D2)^7;

Local F2 = ((A1+B1+C1+D1)+A2+B2+C2+D2)^7;

.end

*** test9.frm

CFunction f,S,R;

Symbol x,N;

Local F = f(x)+f(x^2)+f(x,x+1)+f;

Local G = S(R(3,1,-2),N+1);

Print; 

.end

Index & Vector
*** test10.frm

Index i1,i2,i3;

Vector p1,p2,p3;

Local F = p1(i1)*(p2(i1)+p3(i3))*(p1(i2)+p2(i3));

Print;

.end

*** test11.frm

Index i1;

Vector p1,p2,p3,p4;

Local F = p1(i1)*p3(i1)*p2(i1)*p4(i1);

Print;

.end

*** test12.frm

Symbol x,D;

Index i1=3,i2 = 4,i3=D,i4=0,i5,i6,i7;

Local F = x*d_(i1,i1)

           +x^2*d_(i2,i2)+x^3*d_(i3,i3)

           +x^4*d_(i1,i2)*d_(i2,i1)+x^5*d_(i2,i1)*d_(i1,i2)

           +x^6*d_(i5,i6)*d_(i6,i7)

           +x^7*d_(i4,i4)

           +x^8*d_(i5,i4)*d_(i4,i7);

Print +s;

.end

*** test13.frm

Symbol D;

Index m1=2,m2=2,m3=2,m4=2;

Index n1=3,n2=3,n3=3,n4=3,n5=3,n6=3;

Index r1=D,r2=D,r3=D,r4=D,r5=D,r6=D;

Local F1 = e_(m1,m2)*e_(m3,m4);

Local F2 = e_(m1,m2)*e_(m2,m3);

Local F3 = e_(m1,m2)*e_(m1,m2);

Local G1 = e_(n1,n2,n3)*e_(n4,n5,n6);

Local G2 = e_(n1,n2,n3)*e_(n3,n4,n5);

Local G3 = e_(n1,n2,n3)*e_(n2,n3,n4);

Local G4 = e_(n1,n2,n3)*e_(n1,n2,n3);

Local H1 = e_(r1,r2,r3)*e_(r4,r5,r6);

Local H2 = e_(r1,r2,r3)*e_(r3,r4,r5);

Local H3 = e_(r1,r2,r3)*e_(r2,r3,r4);

Local H4 = e_(r1,r2,r3)*e_(r1,r2,r3);

Contract;

Print;

.end

*** test14.frm

Index i1,i2,i3,i4;

Vector p1,p2;

Local F = e_(i1,i2,i3,i4)*p1(i1)*p2(i2);

Print;

.end

*** test15.frm

Index i1,i2;

Vector p1,p2,p3,p4;

Local F = e_(i1,i2,p1,p2)*e_(i1,i2,p3,p4);

Contract;

Print;

.end


13



Gamma Matrix
*** test16.frm

Index mu1,mu2,mu3,mu4;

Vector p1,p2,p3,p4;

Local F1 = g_(1,mu1)*g_(1,mu2)*g_(1,mu3)*g_(1,mu4);

Local F2 = g_(2,p1)*g_(2,p2)*g_(2,p3)*g_(2,p4);

Print;

.sort

Tracen,1;

Trace4 2;

Print;

.end

*** test17.frm

Vector p1,p2,p3,p4,q1,q2,q3,q4;

Index i1,i2,i3,i4;

Local F = g_(1,p1,i1,7_,p2,i2,7_,p3,i3,7_,p4,i4,7_)

         *g_(2,q1,i1,7_,q2,i2,7_,q3,i3,7_,q4,i4,7_);

Trace4,1;

Trace4,2;

Print;

.end

Pattern Matching
*** test18.frm

Symbol x,y,z,a,b;

CFunction f;

Local F = f(x)+f(y)+f(4)+f(a+b);

id f(z?) = z^2;

Print;

.end

*** test19.frm

Symbol x,y,z,a,b;

CFunction f;

Local F = f(x)*f(y)*f(y)*f(4)*f(a+b)*f(a+b);

id f(z?)*f(z?) = z^2*f(3,z);

Print +s;

.end

*** test20.frm

Symbol a1,a2,a3,a4,a5,a6;

CFunction f,g;

Local F = f(a1,a2,a3,a4)*f(a1,a3,a2,a1,a4,a5,a6);

id  f(?a,a3,?b) = g(?b,a3,?a);

Print +s;

.end

How to do this?


f(x1)*f(x2)*f(x3)*f(x4)*f(x5) => f(x1,x2,x3,x4,x5) 

*** test21.frm

Symbol x1,...,x10;

auto symbol y;

CFunction f;


Local expr = 1

#do i = 1, 10

    *f(x`i')

#enddo

;


repeat id f(?y1)*f(?y2) = f(?y1,?y2);

.sort


print;

.end

*** test22.frm

Symbol x1,...,x10;

auto symbol y;

CFunction f;


Local expr = 1

#do i = 1, 10

    *f(x`i')

#enddo

;


chainin f;

.sort

print;

.end

*** test23.frm

Symbol a1,a2,a3,a4,a5,a6;

CFunction f,g;

Local F = f(a1,a2,a3,a4)*f(a1,a3,a2,a1,a4,a5,a6);


repeat id f(a1?,a2?,?a) = f(a1)*f(a2,?a);

***chainout f;


Print +s;

.end

fibonacci数列
*** test24.frm

Symbol n,x1,x2;

CFunction f,fib;

Local F = f(10,1,1);

repeat;

  id  f(0,?a) = fib(?a);

  id  f(n?,?a,x1?,x2?) = f(n-1,?a,x1,x2,x1+x2);

endrepeat;

Print; 

.end




Table
*** test25.frm

symbol a1,...,a9;

Table,sparse,coefftable(1);


Local expr = (a1+...+a9)^10;


bracket a1;

.sort


fillexpression coefftable = expr(a1);

.sort


printtable coefftable;

.end



C++内部直接调⽤FORM

• Please watch the example.


